SCIENTIA
Series A: *Mathematical Sciences*, Vol. 20 (2011), 93–111
Universidad Técnica Federico Santa María
Valparaíso, Chile
ISSN 0716-8446

The integrals in Gradshteyn and Ryzhik. Part 18: Some automatic proofs

Christoph Koutschan[a] and Victor H. Moll[a]

Abstract. The evaluation of a selection of entries from the table of integrals by Gradshteyn and Ryzhik is presented using the symbolic software package **HolonomicFunctions**.

1. Introduction

The volume [**7**] is one of the most widely used table of integrals. This work, now in its 7-th edition has been edited and amplified several times. The initial work of the authors I. Gradshteyn and I. M. Ryzhik is now supplemented by entries proposed by a large number of researchers.

This paper is part of a project, initiated in [**10**], with the goal of establishing the validity of these formulas and to place them in context. The previous papers in this project contain evaluation of entries in [**7**] by traditional analytical methods. Symbolic languages, mostly *Mathematica*, have so far only been used to check the entries in search for possible errors (e.g., by numerical evaluation). The methodology employed here is different: the computer package **HolonomicFunctions** is employed to deliver *computer-generated proofs* of some entries in [**7**]. The examples are chosen to illustrate the different capabilities of the package. Note that Mathematica (version 7) fails on most of these examples.

2. The class of holonomic functions

The computer algebra methods employed here originate in Zeilberger's holonomic systems approach [**3, 8, 12**]. They can be seen as a generalization of the Almkvist-Zeilberger algorithm [**2**] to integrands that are not necessarily hyperexponential. The basic idea is that of the representation of a function (or a sequence) as solutions of differential (or difference) equations together with some initial conditions. These equations are required to be linear, homogeneous and with polynomial coefficients. It is convenient to present them in operator notation: D_x is used for the (partial)

2000 *Mathematics Subject Classification.* Primary 33.
Key words and phrases. Integrals, automatic proofs.

derivative with respect to x and S_n for the shift in n. The main advantage of this notation is that the differential equations and recurrences under consideration turn into polynomials, which are the basic objects in computer algebra.

Consider first the case of functions of two variables: a continuous one x and a discrete one n. Define $\mathbb{O}$ to be the algebra generated by the operators D_x and S_n with coefficients that are rational functions in x and n. This is a non-commutative algebra and the rules $D_x x = xD_x + 1$ and $S_n n = nS_n + S_n$ must be incorporated into $\mathbb{O}$. Such structures are usually called *Ore algebras*. A similar definition can be made in the case of many continuous and discrete variables.

The operator $P \in \mathbb{O}$ is said to *annihilate* the function f if $P(f) = 0$. For example, $\sin x$ is annihilated by the operator D_x^2+1 and the Fibonacci numbers F_n by $S_n^2-S_n-1$. The latter is nothing but the recurrence $F_n = F_{n-1} + F_{n-2}$ used to define these numbers. Given a function f, the set

$$\operatorname{Ann}_{\mathbb{O}}(f) := \{P \in \mathbb{O} : P(f) = 0\} \tag{2.1}$$

represents the set of all the equations safisfied by f. Naturally, a given operator P may annihilate many different functions. For instance, $D_x^2 + 1$ also annihilates $\cos x$. Thus, $\operatorname{Ann}_{\mathbb{O}}(f)$ does not determine f uniquely: initial conditions must be included. The term *equation* in the present context refers to an equation of the form $P(f) = 0$ with $P \in \mathbb{O}$. Many classical functions, such as rational or algebraic functions, exponentials, logarithms, and some of the trigonometric functions, as well as a multitude of special functions satisfy equations of the type described above.

Observe that $\operatorname{Ann}_{\mathbb{O}}(f)$ is a left ideal in the algebra $\mathbb{O}$, called the *annihilating ideal of* f. Indeed, given $P \in \mathbb{O}$, that represents the equation $P(f) = 0$ satisfied by f, its differentiation yields a new equation for f, represented by D_xP. Similarly, if f depends on the discrete index n and satisfies a linear recurrence $Q(f) = 0$ for $Q \in \mathbb{O}$, then f also satisfies the shifted recurrence, represented by S_nQ.

The annihilating left ideal can be described by a suitable set of generators. The concept of Gröbner bases is then employed to decide the *ideal membership problem*: in our context, to decide whether a function satisfies a given equation. These bases are also used to obtain a unique representation of the residue classes modulo an ideal.

In the holonomic systems approach all manipulations are carried out with these implicit function descriptions, e.g., an algorithm for computing a definite integral requires as input an implicit description of the integrand (viz. the Gröbner basis of an annihilating ideal of the integrand) and will return an implicit description for the integral. The initial values are usually considered afterwards.

In this paper we deal with the so-called *holonomic functions*. Apart from some technical aspects, the most important necessary condition for a function to be holonomic is that there exists an annihilating ideal of dimension zero (a concept that depends on the choice of the underlying algebra $\mathbb{O}$). Equivalently, for each continuous variable x for which D_x belongs to $\mathbb{O}$, there must exist an ordinary differential equation, and similarly, a pure recurrence equation for each discrete variable under consideration. For example, the function $\tan x$ is not holonomic since it it does not satisfy any linear differential equation with polynomial coefficients. Similarly, if f is an arbitrary smooth function, then

$$g(z,m) := \left(\frac{d}{dz}\right)^m f(z) \tag{2.2}$$

is annihilated by the operator $D_z - S_m$. But this single operator is not enough for $g(z,m)$ to be holonomic. In the general situation, there exist neither a pure differential equation (without shifts), nor a pure recurrence (without D_x). However, in special instances, e.g., $f(z) = \sqrt{z}$, the function $g(z,m)$ turns out to be holonomic, being annihilated in this case by the pure operators $2xD_x - m$ and $S_m^2 - x$.

The symbolic framework employed here includes algorithms for *basic arithmetic*, that are referred to as *closure properties*, i.e., given two annihilating left ideals for holonomic functions f and g, respectively, it is possible to compute such an ideal for $f+g$, fg, and $P(f)$, where P is an operator in the underlying Ore algebra. Furthermore, certain substitutions are allowed: an algebraic expression in some continuous variables may be substituted for a continuous variable, and a $\mathbb{Q}$-linear combination of discrete variables may be substituted for a discrete variable. Finally, the definite integral of a holonomic function is again holonomic. Note that quotients and compositions of holonomic functions are not holonomic in general.

The question of deciding if a given function is holonomic is non-trivial. Most of the results are indirect. For instance, Flajolet et al. [**5**] use the fact that a univariate holonomic function has finitely many singularities to illustrate the fact that the generating function of partitions $P(z) = \prod_n (1-z^n)^{-1}$ is non-holonomic. In this same paper, the authors establish conjectures of S. Gerhold [**6**] on the non-holonomicity of the sequences $\log n$ and p_n, the n-th prime number. The closure properties described above indicate that the function $\sin x \, \cos x$ is holonomic whereas $\sin x / \cos x = \tan x$ is not. Likewise, $\sin(\sqrt{1-x^2})$ and F_{2m+k} (with F_n denoting the n-th Fibonacci number) are holonomic, whereas $\cos(\sin x)$ and F_{n^2} are not.

The main tool for computing definite integrals with the holonomic systems approach is a technique called *creative telescoping*. It consists in finding annihilating operators of a special form. For example, in the computation of the definite integral $F = \int_a^b f \, dx$ assume that the integrand f contains additional variables other than x. Then an operator T in the annihilating ideal of f of the form $T = P + D_x Q$ is desired, with the condition that P does neither contain x nor D_x. It is then straight-forward to produce an equation for the integral F:

$$\begin{aligned} 0 &= \int_a^b T(f)\,dx \\ &= \int_a^b P(f)\,dx + \int_a^b \frac{d}{dx} Q(f)\,dx \\ &= P(F) + \big[Q(f)\big]_{x=a}^{x=b}. \end{aligned}$$

The operator P is called the *principal part* or the telescoper, and Q is called the *delta part*. If the summand coming from the delta part does not simplify to zero, the resulting equation can be homogenized. The operator T is usually found by making an ansatz with undetermined coefficients. Using the fact that reduction with the

Gröbner basis yields a unique representation of the remainder, a linear system for the unknowns is obtained by coefficient comparison when equating the remainder to zero. This algorithm has been proposed by Chyzak [**3, 4**]. An algorithm due to Takayama [**11**] uses elimination techniques for computing only the principal part P. It therefore can only be applied if it can be assured a priori that the delta part will vanish; this situation is called *natural boundaries*.

The integrals presented in the rest of the paper illustrate these concepts. These evaluations were obtained using the Mathematica package **HolonomicFunctions**, developed by the first author in [**8**]. It can be downloaded from the webpage

http://www.risc.uni-linz.ac.at/research/combinat/software/

for free, and also a Mathematica notebook containing all examples of this paper, is available there. The commands required for the use of this package are described as they are needed, but more information is provided in [**9**]. The whole paper is organized as a single Mathematica session which we start by loading the package:

In[1]:= << **HolonomicFunctions.m**

HolonomicFunctions package by Christoph Koutschan, RISC-Linz,
Version 1.3 (25.01.2010)
⟶ Type ?HolonomicFunctions for help

3. A first example: The indefinite form of Wallis' integral

The first example considered here deals with the indefinite integral

$$I_n(x) = \int \frac{dx}{(1+x^2)^n}. \tag{3.1}$$

Entry 2.148.3 in [**7**] states the recurrence

$$I_n(x) = \frac{1}{2n-2}\frac{x}{(1+x^2)^{n-1}} + \frac{2n-3}{2n-2}I_{n-1}(x). \tag{3.2}$$

Note that this entry does not provide a closed-form evaluation, but a recursive description. Hence this example is very much in the spirit of the methods described in Section 2. The recursive nature of 2.148.3 becomes even more striking after shifting $n \mapsto n+1$ and rearranging to produce:

$$2n\, I_{n+1}(x) - (2n-1)I_n(x) = \frac{x}{(1+x^2)^n}. \tag{3.3}$$

The package **HolonomicFunctions** is now used to compute this integral directly. The command **Annihilator**[**expr**, **ops**] produces annihilating operators for **expr** with respect to the Ore operators **ops**:

In[2]:= **Annihilator**[**Integrate[1/(1 + x^2)^n, x], S[n]**]

Out[2]= $\{(2nx^2+2n+2x^2+2)S_n^2 + (-2nx^2-4n-x^2-1)S_n + (2n-1)\}$

In order to verify that this result agrees with (3.2) it is required to produce a homogeneous version of the latter. This is achieved by left multiplying by an annihilating operator of the inhomogeneous part of (3.3). The expression $x(1+x^2)^{-n}$ is clearly annihilated by $(x^2+1)S_n - 1$ and hence we obtain

$$\big((x^2+1)S_n - 1\big)\cdot\big(2nS_n - (2n-1)\big) =$$
$$2(n+1)(x^2+1)S_n^2 + (-2nx^2 - 4n - x^2 - 1)S_n + (2n-1),$$

matching the operator computed by the program. An alternative procedure is to employ an option that produces inhomogeneous relations:

In[3]:= **Annihilator[Integrate[1/(1 + x^2)^n, x], S[n], Inhomogeneous → True]**

Out[3]= $\left\{\{2nS_n + (1-2n)\}, \left\{-x\left(x^2+1\right)^{-n}\right\}\right\}$

The first part of the result is the operator to be applied to the integral. Adding the second part and equating to zero yields (3.3).

4. A differential equation for hypergeometric functions in two variables

The second example appears as entry **9.181.1** in [**7**]. It concerns differential equations for the hypergeometric function

$$F_1(\alpha,\beta,\beta',\gamma;x,y) = \sum_{m=0}^{\infty}\sum_{n=0}^{\infty}\frac{x^m y^n (\beta)_m(\beta')_n(\alpha)_{m+n}}{m!n!(\gamma)_{m+n}} \tag{4.1}$$

for $|x|<1$ and $|y|<1$. This is defined in **9.180.1**.

Again, the nature of this example is that no closed form is desired. The result is a system of partial differential equations. These equations are now derived completely automatically from (4.1). To achieve this, Takayama's algorithm is employed. Details about this integration algorithm are provided in [**8**].

First observe that both sums have natural boundaries, therefore Takayama's algorithm may be applied. The input is an annihilating ideal for the summand. This is obtained with the command **Annihilator** described in the previous section. The computation is direct since the summand is hypergeometric and hyperexponential in all variables. The first step

In[4]:= **ann = Annihilator[Pochhammer[α, m + n] Pochhammer[β, m] Pochhammer[b, n]/(Pochhammer[γ, m + n] m! n!) x^m y^n, {S[m], S[n], Der[x], Der[y]}]**

Out[4]= $\{yD_y - n, xD_x - m,$
$(mn + m + n^2 + n\gamma + n + \gamma)S_n + (-bmy - bny - by\alpha - mny - n^2y - ny\alpha),$
$(m^2 + mn + m\gamma + m + n + \gamma)S_m + (-m^2x - mnx - mx\alpha - mx\beta - nx\beta - x\alpha\beta)\}$

finds the annihilating ideal for the summand in the hypergeometric function (where the parameter β' has been replaced by b). The second step

In[5]:= **pde = Takayama[ann, {m, n}]**

Out[5]= $\{(xy^2 - xy - y^3 + y^2)D_y^2 + (bx^2 - bx)D_x$
$+ (bxy - by^2 + xy\alpha - xy\beta + xy + x\beta - x\gamma - y^2\alpha - y^2 + y\gamma)D_y + (bx\alpha - by\alpha),$
$(x-y)D_xD_y - bD_x + \beta D_y,$
$(x^3 - x^2y - x^2 + xy)D_x^2 + (bxy - by + x^2\alpha + x^2\beta + x^2 - xy\alpha - xy\beta - xy - x\gamma + y\gamma)D_x$
$+ (y\beta - y^2\beta)D_y + (x\alpha\beta - y\alpha\beta)\}$

performs the double summation and computes a Gröbner basis for the ideal containing all the differential equations satisfied by the series F_1. Observe that the two equations given in **9.181.1** are not among these generators. Thus, it is required to verify whether the differential equations given in [**7**] are members of the ideal. This is achieved by reducing them with the Gröbner basis and check whether the remainder is zero; the necessary command is **OreReduce**.

The program allows for a better result. The desired equations can be produced automatically by observing that the first is free of β' and second does not involve β. The elimination of a parameter can be either done by another Gröbner basis computation (i.e., elimination by rewriting) or by using the command **FindRelation** that performs elimination by ansatz.

The command **FindRelation**[**ann**, **opts**] computes relations in the annihilating ideal **ann** specified by the options **opts**. In this case, the option **Eliminate** forces the coefficients to be free of the given variables.

In[6]:= **FindRelation[pde, Eliminate → $\boldsymbol{\beta}$]**

Out[6]= $\{(xy-x)D_xD_y+(y^2-y)D_y^2+bxD_x+(by+y\alpha+y-\gamma)D_y+b\alpha\}$

Alternatively, the command **OreGroebnerBasis**[$\{P_1,\dots,P_k\}$, **alg**] translates the operators $P_1,\dots,P_k$ into the Ore algebra **alg** and then computes their left Gröbner basis.

In[7]:= **OreGroebnerBasis[pde, OreAlgebra[*b*, Der[*y*], Der[*x*]],
MonomialOrder → EliminationOrder[1]] // First**

Out[7]= $(xy-y)D_yD_x+(x^2-x)D_x^2+y\beta D_y+(x\alpha+x\beta+x-\gamma)D_x+\alpha\beta$

This is precisely the form in which these differential equations are given in [**7**].

5. An integral involving Chebyshev polynomials

The symbolic algorithms implemented in **HolonomicFunctions** do not provide closed-form expressions for definite integrals. Their main use in the evaluation of integrals is based on the fact that, in many examples, it is possible to produce a computer-generated proof of the stated identities. In this sense, *both sides* of an identity are required. The program yields an automatic proof of its validity. The example presented here appears as entry **7.349** in [**7**]:

$$\int_{-1}^{1}(1-x^2)^{-1/2}\,T_n(1-x^2y)\,dx=\frac{\pi}{2}\big(P_{n-1}(1-y)+P_n(1-y)\big). \tag{5.1}$$

Here $T_n(x)$ is the Chebyshev polynomial of the first kind defined by

$$T_n(x)=\cos(n\arccos x)$$

and the answer contains the Legendre polynomial $P_n(x)$ defined by

$$P_n(x)=\frac{1}{2^n n!}\frac{d^n}{dx^n}\left(x^2-1\right)^n.$$

This simple example is chosen to describe in more detail what the software does in the background.

The starting point is the computation of an annihilating ideal for the integrand in (5.1). The integrand is referred as $f(n,x,y)$. For this purpose, recall the recurrence

$$T_{n+2}(z) - 2zT_{n+1}(z) + T_n(z) = 0 \tag{5.2}$$

and the differential equation

$$(z^2-1)T_n''(z) + zT_n'(z) - n^2T_n(z) = 0 \tag{5.3}$$

for the Chebyshev polynomials. These basic relations are stored in a database that the software can access. An easy argument shows that f satisfies the recurrence (5.2) if z is replaced by $1-x^2y$. Observe that the factor $(1-x^2)^{-1/2}$ (which is constant with respect to n) does not change the recurrence. The same substitution is performed in (5.3) and considering $T_n(1-x^2y)$ as a function in y yields

$$\frac{(1-x^2y)^2-1}{x^4}\frac{\partial^2}{\partial y^2}T_n(1-x^2y) + \frac{1-x^2y}{-x^2}\frac{\partial}{\partial y}T_n(1-x^2y) - n^2T_n(1-x^2y) = 0.$$

Multiplication by x^2 produces the annihilating operator

$$(x^2y^2-2y)D_y^2 + (x^2y-1)D_y - n^2x^2$$

for the integrand f. Once again, the square root term does not play a role since it is free of the variable y. Finally, observe that

$$\begin{aligned}\frac{df}{dx} &= \frac{-2xy}{\sqrt{1-x^2}}T_n'(1-x^2y) + \frac{x}{(1-x^2)^{3/2}}T_n(1-x^2y)\\ \frac{df}{dy} &= \frac{-x^2}{\sqrt{1-x^2}}T_n'(1-x^2y)\end{aligned}$$

giving rise to the operator

$$xD_x - 2yD_y - \frac{x^2}{1-x^2}$$

which also annihilates f.

Similar operations, with less ad hoc tricks but more algorithmic steps, are performed by typing the command

In[8]:= $\mathbf{Annihilator}\big[\mathbf{ChebyshevT}[\boldsymbol{n}, \mathbf{1}-\boldsymbol{x}^{\mathbf{2}}\boldsymbol{y}]/\mathbf{Sqrt}[\mathbf{1}-\boldsymbol{x}^{\mathbf{2}}],\ \{\mathbf{S}[\boldsymbol{n}], \mathbf{Der}[\boldsymbol{x}], \mathbf{Der}[\boldsymbol{y}]\}\big]$

Out[8]= $\big\{(x^3-x)D_x + (2y-2x^2y)D_y + x^2,$
$nS_n + (x^2y^2-2y)D_y + (nx^2y-n),$
$(x^2y^2-2y)D_y^2 + (x^2y-1)D_y - n^2x^2\big\}$

Since the command **Annihilator** always returns a Gröbner basis, the above operators differ slightly from the ones that were derived by hand. But these can be obtained by simple combining and rewriting.

The next step in the evaluation of the integral (5.1) employs a new command: **CreativeTelescoping**[f, Der[x], **ops**] computes a set of operators $P_i + D_xQ_i$ of the form described in Section 2. These operators annihilate the function f and they are chosen in a way that the principal parts P_i form a Gröbner basis in the Ore algebra generated by **ops**.

In[9]:= **CreativeTelescoping[ChebyshevT[n, 1 − x^2 y]/Sqrt[1 − x^2], Der[x], {S[n], Der[y]}]**

Out[9]= $$\Big\{\big\{(2n^2+2n)S_n+(2ny^2-4ny+y^2-2y)D_y+(2n^2y-2n^2+ny-2n),\ (y^2-2y)D_y^2+(y-2)D_y-n^2\big\},\ \Big\{\frac{y\left(x^4y-x^2y-2x^2+2\right)}{x}D_y+y\left(nx^3-nx\right),\ \frac{x^2-1}{x}D_y\Big\}\Big\}$$

The first list contains two principal parts P_1 and P_2, and the second list contains the corresponding delta parts Q_1 and Q_2. It is easily verified that the latter do not contribute:

In[10]:= **ApplyOreOperator[Last[%], ChebyshevT[n, 1−x^2y]/Sqrt[1−x^2]] // Simplify**

Out[10]= $$\Big\{-nx\sqrt{1-x^2}y\left(\text{ChebyshevT}\left[n,1-x^2y\right]+\left(2-x^2y\right)\text{ChebyshevU}\left[n-1,1-x^2y\right]\right),\ nx\sqrt{1-x^2}\text{ChebyshevU}\left[n-1,1-x^2y\right]\Big\}$$

In[11]:= **(Limit[#, x → 1] − Limit[#, x → −1])& /@ %**

Out[11]= $\{0,0\}$

It follows that P_1 and P_2 annihilate the integral. All the previous steps are performed in the background by typing

In[12]:= **Annihilator[Integrate[ChebyshevT[n, 1 − x^2y]/Sqrt[1 − x^2], {x, −1, 1}], {S[n], Der[y]}]**

Out[12]= $$\big\{(2n^2+2n)S_n+(2ny^2-4ny+y^2-2y)D_y+(2n^2y-2n^2+ny-2n),\ (y^2-2y)D_y^2+(y-2)D_y-n^2\big\}$$

The next step is the computation of an annihilating ideal for the right-hand side of (5.1). In this process the fact that the sum of the two Legendre polynomials can be written as $Q(P_{n-1}(1-y))$ with $Q=S_n+1$, is employed. This observation produces simpler results (see also equation (10.3) for a more detailed discussion on this issue).

In[13]:= **rhs = Annihilator[ApplyOreOperator[S[n] + 1, LegendreP[n − 1, 1 − y]], {S[n], Der[y]}]**

Out[13]= $$\big\{(2n^2+2n)S_n+(2ny^2-4ny+y^2-2y)D_y+(2n^2y-2n^2+ny-2n),\ (y^2-2y)D_y^2+(y-2)D_y-n^2\big\}$$

This produces the same annihilating ideal for the right-hand side as the one produced for the left-hand side. The desired identity is now obtained by comparing some initial values. The necessary cases can be read off from the shape of the Gröbner basis. They correspond to the monomials that lie under the stairs which is formed by the leading monomials of this basis. The required command is:

In[14]:= **UnderTheStaircase[rhs]**

Out[14]= $\{1, D_y\}$

In other words, we identify those instances (shifts and derivatives) of the function that cannot be reduced by using the annihilating operators in **rhs**. Hence it is required to check whether $L_0(0)=R_0(0)$ and $L_0'(0)=R_0'(0)$. Here $L_n(y)$ and $R_n(y)$ denote the left and right side of (5.1). This is left as an exercise to the reader.

6. An integral involving a hypergeometric function

The example of this section appears as entry **7.512.5** in [**7**]: for $\operatorname{Re} r > 0$, $\operatorname{Re} s > 0$, and $\operatorname{Re}(c + s - a - b) > 0$,

$$\int_0^1 x^{r-1}(1-x)^{s-1}{}_2F_1(a,b;c;x)\,dx = \frac{\Gamma(r)\Gamma(s){}_3F_2(a,b,r;c,r+s;1)}{\Gamma(r+s)} \tag{6.1}$$

where

$$ {}_pF_q(a_1,\ldots,a_p;b_1,\ldots,b_q;x) := \sum_{k=0}^{\infty} \frac{(a_1)_k\cdots(a_p)_k}{(b_1)_k\cdots(b_q)_k k!}x^k \tag{6.2}$$

is the classical hypergeometric series.

The strategy of the proof is to compute a system of recurrences for each side of the identity. These recurrences then reduce the problem to checking finitely many initial values. For this purpose, the parameters a, b, c, r, s are assumed to be integers. The first step is to compute an annihilating ideal of the left-hand side:

In[15]:= **lhs = Annihilator[Integrate[x^(r − 1) (1 − x)^(s − 1) Hypergeometric2F1[a, b, c, x], {x, 0, 1}], {S[a], S[b], S[c], S[r], S[s]}, Assumptions → s > 1]**

Out[15]=
$$\begin{aligned}\{&S_r + S_s - 1,\\
&(abc - abr - ac^2 + acr - bc^2 + bcr + c^3 - c^2r)S_c\\
&\quad + (-abc + acr + acs + bcr + bcs - cr^2 - 2crs - cs^2)S_s\\
&\quad + (ac^2 - acr - acs + bc^2 - bcr - bcs - c^3 + c^2r + crs + cs^2),\\
&(-ab-b^2+bc+bs-b)S_b+(ab-ar-as-br-bs+r^2+2rs+s^2)S_s+(as+b^2-bc+b-rs-s^2),\\
&(-a^2-ab+ac+as-a)S_a+(ab-ar-as-br-bs+r^2+2rs+s^2)S_s+(a^2-ac+a+bs-rs-s^2),\\
&(ab - ar - as - a - br - bs - b + r^2 + 2rs + 2r + s^2 + 2s + 1)S_s^2\\
&\quad + (-ab + ar + 2as + a + br + 2bs + b - cr - cs - 2rs - r - 2s^2 - 2s - 1)S_s\\
&\quad + (-as - bs + cs + s^2)\}\end{aligned}$$

The restriction $s > 1$ is a technicality: the automatic simplification does not succeed if $s \geqslant 1$ is imposed. The special case $s = 1$ can be done separately in the very same manner. Annihilating operators for the right side are obtained in a similar fashion:

In[16]:= **rhs = Annihilator[Gamma[r] Gamma[s]/Gamma[$r + s$] HypergeometricPFQ[{a, b, r}, {$c, r + s$}, 1], {S[a], S[b], S[c], S[r], S[s]}]**

DFiniteSubstitute::divzero : Division by zero happened during algebraic substitution (caused by a singularity in the original annihilator). The result is not guaranteed to be correct. Please check whether this substitution and the output given below make sense.

Out[16]=
$$\begin{aligned}\{&S_r + S_s - 1,\\
&(-abc + abr + ac^2 - acr + bc^2 - bcr - c^3 + c^2r)S_c\\
&\quad + (abc - acr - acs - bcr - bcs + cr^2 + 2crs + cs^2)S_s\\
&\quad + (-ac^2 + acr + acs - bc^2 + bcr + bcs + c^3 - c^2r - crs - cs^2),\\
&(ab+b^2-bc-bs+b)S_b+(-ab+ar+as+br+bs-r^2-2rs-s^2)S_s+(-as-b^2+bc-b+rs+s^2),\\
&(a^2 + ab - ac - as + a)S_a + (-ab + ar + as + br + bs - r^2 - 2rs - s^2)S_s\\
&\quad + (-a^2 + ac - a - bs + rs + s^2),\\
&(ab - ar - as - a - br - bs - b + r^2 + 2rs + 2r + s^2 + 2s + 1)S_s^2\end{aligned}$$

$$+ (-ab + ar + 2as + a + br + 2bs + b - cr - cs - 2rs - r - 2s^2 - 2s - 1)S_s$$
$$+ (-as - bs + cs + s^2)\}$$

This unfavorable warning comes from the factor $(1-x)$ that appears in the leading coefficient of the hypergeometric differential equation, making $x = 1$ a singular value. To avoid this issue, the necessary relations are derived by a different approach. Applying Takayama's algorithm to (6.2) yields:

```
In[17]:= smnd = Pochhammer[a, k] Pochhammer[b, k] Pochhammer[r, k]
             /Pochhammer[c, k]/Pochhammer[r + s, k]/k!;
In[18]:= tak = Takayama[Annihilator[smnd, {S[k], S[a], S[b], S[c], S[r], S[s]}], {k}];
In[19]:= rhs2 = DFiniteTimes[Annihilator[Gamma[r] Gamma[s]/Gamma[r + s],
             {S[a], S[b], S[c], S[r], S[s]}], tak];
In[20]:= GBEqual[rhs, rhs2]
Out[20]= True
```

The lengthy output has been suppressed, but the last line shows that the annihilating ideal is identical to the one obtained before. Therefore, both sides of identity (6.1) are annihilated by the same operator ideal.

The next step is to determine the initial values required to complete the proof. A range of parameters is fixed first, say $a \geqslant 0, b \geqslant 0, c \geqslant 1, r \geqslant 1, s \geqslant 1$. Taking the recurrences as the defining equations, it is now required to find the values needed to determine a multivariate sequence uniquely. In the univariate case, this corresponds to the first d values when d is the order of the recurrence. The multivariate analog are all monomials that lie under the stairs of the Gröbner basis.

Morever, as in the univariate case the vanishing of some leading coefficient in the recurrences has to be investigated. In the univariate case, this question reduces to finding the nonnegative integer roots of the leading coefficient. In the multivariate case this analysis is more intriguing. It can even happen that there are infinitely many such singular points. Therefore a command that automatically determines all these critical points has been implemented:

```
In[21]:= sing = AnnihilatorSingularities[lhs, {0, 0, 1, 1, 1},
             Assumptions → c + s − a − b > 0]
Out[21]= {{{a → 0, b → 0, c → 1, r → 1, s → 1}, True},
          {{a → 0, b → 0, c → 1, r → 1, s → 2}, True},
          {{a → 0, b → 0, c → 2, r → 1, s → 1}, True},
          {{a → 0, b → 0, c → 2, r → 1, s → 2}, True},
          {{a → 0, b → 1, c → 1, r → 1, s → 1}, True},
          {{a → 0, b → 1, c → 1, r → 1, s → 2}, True},
          {{a → 0, b → 1, c → 2, r → 1, s → 1}, True},
          {{a → 0, b → 1, c → 2, r → 1, s → 2}, True},
          {{a → 1, b → 0, c → 1, r → 1, s → 1}, True},
          {{a → 1, b → 0, c → 1, r → 1, s → 2}, True},
          {{a → 1, b → 0, c → 2, r → 1, s → 1}, True},
          {{a → 1, b → 0, c → 2, r → 1, s → 2}, True},
          {{a → 1, b → 1, c → 1, r → 1, s → 2}, True},
          {{a → 1, b → 1, c → 1, r → 1, s → 3}, True},
```

$\{\{a \to 1, b \to 1, c \to 2, r \to 1, s \to 1\}, \text{True}\},$
$\{\{a \to 1, b \to 1, c \to 2, r \to 1, s \to 2\}, \text{True}\}\}$

The last entry in each case states the condition under which the corresponding points are singular. It is seen that there are many more such points than the staircase of the Gröbner basis would indicate:

In[22]:= **UnderTheStaircase[lhs]**

Out[22]= $\{1, S_s\}$

It is now routine to check the identity (6.1) for the above 16 cases, since in most of them the hypergeometric function in the integral reduces to a simple polynomial:

In[23]:= **(MyInt[*x*^(*r* − 1) (1 − *x*)^(*s* − 1) Hypergeometric2F1[*a*, *b*, *c*, *x*], {*x*, 0, 1}] ==**
Gamma[*r*] Gamma[*s*]/Gamma[*r* + *s*]
HypergeometricPFQ[{*a*, *b*, *r*}, {*c*, *r* + *s*}, 1])
/. (First /@ sing) /. MyInt → Integrate

Out[23]= {True, True, True, True, True, True, True, True, True, True, True, True, True,
True, True, True}

Hence entry **7.512.5** in [**7**] has been verified.

7. An integral involving Gegenbauer polynomials

The next identity appears as entry **7.322** in [**7**]: it states that

$$\int_0^{2a} e^{-bx}(x(2a-x))^{\nu-\frac{1}{2}} C_n^{(\nu)}\left(\frac{x}{a}-1\right)\,dx = \frac{\pi(-1)^n e^{-ab}\left(\frac{a}{2b}\right)^{\nu}\Gamma(n+2\nu)I_{n+\nu}(ab)}{n!\,\Gamma(\nu)}$$

where $C_n^{(\nu)}(x)$ denote the Gegenbauer polynomials and $I_\nu(x)$ the modified Bessel function of the first kind. The former special function is defined via the generating function

$$(1-2x\alpha+\alpha^2)^{-\nu} = \sum_{n=0}^{\infty} C_n^{(\nu)}(x)\alpha^n \tag{7.1}$$

and the latter is

$$I_\nu(x) = \sum_{k=0}^{\infty} \frac{1}{k!\,\Gamma(\nu+k+1)}\left(\frac{x}{2}\right)^{\nu+2k}. \tag{7.2}$$

The computation of the annihilating ideal for the integral requires some human intervention: the problem is that in this instance the inhomogeneous part cannot be evaluated automatically. Hence the option **Inhomogeneous→True** once again is used and the simplifications are done "by hand". It turns out that all inhomogeneous parts evaluate to 0.

In[24]:= **{lhs, inh} = Annihilator[Integrate[**
((*x*(2*a* − *x*))^(*ν* − 1/2) GegenbauerC[*n*, *ν*, *x*/*a* − 1])/E^(*bx*), {*x*, 0, 2*a*}],
{Der[*a*], Der[*b*], S[*n*], S[*ν*]},
Assumptions → *ν* ⩾ 1,
Inhomogeneous → True];

In[25]:= **Simplify[ReleaseHold[inh /. Limit → myLimit]] /. myLimit → Limit,**
Assumptions → *ν* ⩾ 1] // Simplify

Out[25]= {0, 0, 0, 0}

The right-hand side can be handled in completely automatic fashion and we obtain exactly the same differential-difference operators as for the other side:

In[26]:= $\mathbf{rhs = Annihilator[}$
$\mathbf{(-1)^n Pi\,Gamma[2\nu+n]/n!/Gamma[\nu](a/(2b))^\nu BesselI[\nu+n,ab]/E\hat{}(ab),}$
$\mathbf{\{Der[a],Der[b],S[n],S[\nu]\}]}$

Out[26]= $\{(an^2+2an\nu+2an+2a\nu+a)S_n+2b\nu S_\nu,$
$(bn^2+4bn\nu+bn+4b\nu^2+2b\nu)D_b-2b^2\nu S_\nu$
$\quad +(abn^2+4abn\nu+abn+4ab\nu^2+2ab\nu-n^3-4n^2\nu-n^2-4n\nu^2-2n\nu),$
$(an^2+4an\nu+an+4a\nu^2+2a\nu)D_a-2b^2\nu S_\nu$
$\quad +(abn^2+4abn\nu+abn+4ab\nu^2+2ab\nu-n^3-6n^2\nu-n^2-12n\nu^2-4n\nu-8\nu^3-4\nu^2),$
$(4b^2\nu^2+4b^2\nu)S_\nu^2+(4n^3\nu+20n^2\nu^2+24n^2\nu+32n\nu^3+76n\nu^2+44n\nu+16\nu^4+56\nu^3$
$\quad +64\nu^2+24\nu)S_\nu+(-a^2n^4-8a^2n^3\nu-6a^2n^3-24a^2n^2\nu^2-36a^2n^2\nu-11a^2n^2$
$\quad -32a^2n\nu^3-72a^2n\nu^2-44a^2n\nu-6a^2n-16a^2\nu^4-48a^2\nu^3-44a^2\nu^2-12a^2\nu)\}$

In[27]:= **GBEqual[lhs, rhs]**

Out[27]= True

This is already a strong indication that the identity is correct, but the initial values have to be compared. There are two monomials under the stairs of the Gröbner basis **lhs** (or **rhs** which is the same):

In[28]:= **UnderTheStaircase[lhs]**

Out[28]= $\{1, S_\nu\}$

Hence the initial values for $\nu=0$ and $\nu=1$ have to be compared. The values for a, b, and n can be prescribed, since the corresponding operators D_a, D_b, and S_n had been included in the algebra:

In[29]:= $\mathbf{(x(2a-x))\hat{}(\nu-1/2)\,GegenbauerC[n,\nu,x/a-1]/E\hat{}(bx)}$
$\mathbf{/.\ \{a\to 1, b\to 1, n\to 1\}\ /.\ \{\{\nu\to 0\},\{\nu\to 1\}\}}$

Out[29]= $\left\{0, 2(x-1)\sqrt{(2-x)x}e^{-x}\right\}$

In[30]:= $\mathbf{Integrate[\%,\{x,0,2\}]}$

Out[30]= $\left\{0, -\frac{2\pi \mathrm{BesselI}[2,1]}{e}\right\}$

In[31]:= $\mathbf{(-1)^n Pi\,Gamma[2\nu+n]/n!/Gamma[\nu]\ (a/(2b))^\nu BesselI[\nu+n,ab]/E\hat{}(ab)}$
$\mathbf{/.\ \{a\to 1, b\to 1, n\to 1\}\ /.\ \{\{\nu\to 0\},\{\nu\to 1\}\}}$

Out[31]= $\left\{0, -\frac{\pi \mathrm{BesselI}[2,1]}{e}\right\}$

Obviously, the right-hand side has a factor 2 missing. *Hence a misprint in the book has been found!*

8. The product of two Bessel functions

For $a>b>0$ and $\mathrm{Re}\,(m+n)>-1$, formula **6.512.1** states that

$$\int_0^\infty J_m(ax)J_n(bx)\,dx = \frac{b^n}{a^{n+1}}\frac{\Gamma\left(\frac{m+n+1}{2}\right)}{\Gamma(n+1)\Gamma\left(\frac{m-n+1}{2}\right)}\,{}_2F_1\left(\tfrac{m+n+1}{2},\tfrac{n-m+1}{2};n+1;\tfrac{b^2}{a^2}\right)$$

where $J_n(x)$ denotes the Bessel function. This classical special function is defined by the series

$$J_\nu(z) = \left(\frac{z}{2}\right)^\nu \sum_{k=0}^{\infty} \frac{(-1)^k}{k!\,\Gamma(\nu+k+1)} \left(\frac{z}{2}\right)^{2k}. \tag{8.1}$$

Some problems appeared in the computation of an annihilating ideal for the left-hand side, namely the software complains that it cannot evaluate some delta part. To obtain further details about these problems, the creative telescoping relations were explicitly computed (the delta parts are shown in the output):

In[32]:= $\{\mathbf{ann}, \mathbf{delta}\} = \mathbf{CreativeTelescoping}\big[\mathbf{BesselJ}[\boldsymbol{m}, \boldsymbol{ax}]\, \mathbf{BesselJ}[\boldsymbol{n}, \boldsymbol{bx}],\ \mathbf{Der}[\boldsymbol{x}],$
$\{\mathbf{Der}[\boldsymbol{a}], \mathbf{Der}[\boldsymbol{b}], \mathbf{S}[\boldsymbol{m}], \mathbf{S}[\boldsymbol{n}]\}\big];$

In[33]:= **delta**

Out[33]= $\big\{-x,\ 2bx(an+a)S_mS_n - 2b(n+1)(m+n+1)S_n + 2b^2(nx+x),$
$abxS_mS_n + b^2x,\ -2ax(bm+b)S_mS_n + 2a(m+1)(m+n+1)S_m - 2\left(b^2mx + b^2x\right),$
$-abxS_m + b^2xS_n,\ b^2xS_m - abxS_n,\ ab^2x^2S_m - b^3x^2S_n - b^2(mx - nx - x)\big\}$

The first delta part already reveals the difficulties involved: according to the derivation in Section 2 the boundary condition to evaluate is

$$\big[xJ_m(ax)J_n(bx)\big]_{x=0}^{x=\infty}.$$

The Bessel function $J_n(x)$ is asymptotically equivalent to $\sqrt{2/(\pi x)}$ as $x \to \infty$ (see [**1**, 9.2.1]). Therefore the limit in the first delta part does not exist. Moreover, some other delta parts involve x^2, which makes the situation even worse.

One way to overcome these difficulties consists in going back to the roots of the holonomic systems approach. In his original paper [**12**], Zeilberger suggested to find an operator whose coefficients are completely free of the integration variable x. This is more than necessary, since usually it does not harm if x occurs in the delta part. Once such an operator is found, it is immediate to rewrite it into the form $P + D_xQ$. This method was called the "slow algorithm" by Zeilberger himself, and this points to the reason why it is rarely used in practice. In the example described here, this technique could be useful, since the occurrence of x in the delta part is exactly the problem encountered. It is also desired to have no derivatives with respect to a or b in the delta parts, since they cause the same difficulties. Such operators can be found by means of the command **FindRelation**, where the first condition can be encoded by the option **Eliminate**, and the second by the option **Support**: all monomials up to total degree 2 are given, but D_aD_x and D_bD_x are omitted:

In[34]:= $\mathbf{ops} = \{\mathbf{Der}[\boldsymbol{x}], \mathbf{Der}[\boldsymbol{a}], \mathbf{Der}[\boldsymbol{b}], \mathbf{S}[\boldsymbol{m}], \mathbf{S}[\boldsymbol{n}]\};$

In[35]:= $\mathbf{supp} = \mathbf{Complement}[\mathbf{Join}[\{\mathbf{1}\}, \mathbf{ops}, \mathbf{Flatten}[\mathbf{Outer}[\mathbf{Times}, \mathbf{ops}, \mathbf{ops}]]],$
$\{\mathbf{Der}[\boldsymbol{x}]\,\mathbf{Der}[\boldsymbol{a}],\ \mathbf{Der}[\boldsymbol{x}]\mathbf{Der}[\boldsymbol{b}]\}];$

In[36]:= $\mathbf{rels} = \mathbf{FindRelation}\big[\mathbf{Annihilator}[\mathbf{BesselJ}[\boldsymbol{m}, \boldsymbol{ax}]\, \mathbf{BesselJ}[\boldsymbol{n}, \boldsymbol{bx}], \mathbf{ops}],$
$\mathbf{Eliminate} \to \boldsymbol{x},\ \mathbf{Support} \to \mathbf{supp}\big]$

Out[36]= $\big\{-ab^2mD_aS_m + (a^2bn + a^2b)D_aS_n + (a^2(-b) - a^2bn)D_bS_m + ab^2mD_bS_n$
$+ (a^2n^2 + a^2n - b^2m^2 - b^2m)S_m,$
$(2m+2)D_xS_m + (am - an + a)S_m^2 + (2bm + 2b)S_mS_n + (a(-m) - an - a),$

$$(2n+2)D_xS_n + (2an+2a)S_mS_n + (b(-m)+bn+b)S_n^2 + (b(-m)-bn-b),$$
$$-abD_aS_n + abD_bS_m - anS_m + bmS_n,$$
$$-a^2b^2D_a^2 + a^2b^2D_b^2 - ab^2D_a + a^2bD_b + (b^2m^2 - a^2n^2)\}$$

The principal and delta parts of these 5 relations have to be separated manually. Observe that D_x is not invertible in the underlying Ore algebra. Therefore this operation has to be performed on the level of Mathematica expressions:

In[37]:= **pps = OrePolynomialSubstitute[rels, {Der[*x*] → 0}];**

In[38]:= **deltas = Together[(Normal /@ (rels − pps))/Der[*x*]]**

Out[38]= $\{0,\ 2(m+1)\mathrm{S}[m],\ 2(n+1)\mathrm{S}[n],\ 0,\ 0\}$

Now all the inhomogeneous parts vanish: the limits for $x \to \infty$ as well as the evaluations at $x = 0$. The latter is true because at least one of the orders of the two Bessel functions becomes $\geqslant 1$ (recall that we impose $m \geqslant 0$ and $n \geqslant 0$). Hence the principal parts annihilate the integral. In order to compare them against the right-hand side, the Gröbner basis of the ideal generated by them is computed:

In[39]:= **lhs = OreGroebnerBasis[pps, OreAlgebra[Der[*a*], Der[*b*], S[*m*], S[*n*]], MonomialOrder → DegreeLexicographic]**

Out[39]= $\{aD_a + bD_b + 1,$
$$(b^2m^2 - b^2n^2 - 2b^2n - b^2)S_n^2 + (2a^2bn + 2a^2b - 2b^3n - 2b^3)D_b + (-2a^2n^2 - 2a^2n + b^2m^2 + b^2n^2 - b^2),$$
$$(-abm - abn - ab)S_mS_n + (b^3 - a^2b)D_b + (a^2n + b^2m + b^2),$$
$$(a^2(-m^2) - 2a^2m + a^2n^2 - a^2)S_m^2 + (2a^2bm + 2a^2b - 2b^3m - 2b^3)D_b + (a^2m^2 + 2a^2m + a^2n^2 + a^2 - 2b^2m^2 - 4b^2m - 2b^2),$$
$$(a^2b - b^3)D_bS_n + (abn - abm)S_m + (a^2n + a^2 - b^2m - b^2)S_n,$$
$$(a^2b - b^3)D_bS_m + (b^2m - a^2n)S_m + (abm - abn)S_n,$$
$$(b^4 - a^2b^2)D_b^2 + (3b^3 - a^2b)D_b + (a^2n^2 + b^2(-m^2) + b^2)\}$$

In[40]:= **rhs = Annihilator[**
$b^n\, a^{-n-1}$Gamma[(*m* + *n* + 1)/2]/Gamma[*n* + 1]/Gamma[(*m* − *n* + 1)/2]
Hypergeometric2F1[(*n* + *m* + 1)/2, (*n* − *m* + 1)/2, *n* + 1, b^2/a^2],
{Der[*a*], Der[*b*], S[*m*], S[*n*]}];

In[41]:= **GBEqual[lhs, rhs]**

Out[41]= True

The proof is completed by checking four initial values. This is left to the reader.

9. An example involving parabolic cylinder functions

The first entry in section **3.953** states that

$$\int_0^\infty x^{\mu-1}e^{-\gamma x-\beta x^2}\sin(ax)\,dx = -\frac{i}{2(2\beta)^{\mu/2}}\exp\left(\frac{\gamma^2-a^2}{8\beta}\right)\Gamma(\mu)$$
$$\times\left\{\exp\left(-\frac{ia\gamma}{4\beta}\right)D_{-\mu}\left(\frac{\gamma-ia}{\sqrt{2\beta}}\right) - \exp\left(\frac{ia\gamma}{4\beta}\right)D_{-\mu}\left(\frac{\gamma+ia}{\sqrt{2\beta}}\right)\right\}$$

for $\operatorname{Re}\mu > -1$, $\operatorname{Re}\beta > 0$, and $a > 0$. The symbol $D_s(z)$ denotes the parabolic cylinder function defined by

$$D_s(z) := 2^{s/2}\sqrt{\pi}e^{-\frac{z^2}{4}}\left(\frac{1}{\Gamma\left(\frac{1-s}{2}\right)}\,{}_1F_1\left(-\frac{s}{2},\frac{1}{2},\frac{z^2}{2}\right) - \frac{\sqrt{2}z}{\Gamma\left(-\frac{s}{2}\right)}\,{}_1F_1\left(\frac{1-s}{2},\frac{3}{2},\frac{z^2}{2}\right)\right).$$

Alternatively, this function is defined as a certain solution of the differential equation

$$4y''(x) + \left(4s - x^2 + 2\right)y(x) = 0.$$

For convenience of typing, the greek letters are replaced by roman ones ($\beta = b$, $\gamma = c$, $\mu = m$), and the complicated right-hand side is stored as an extra variable.

```
In[42]:= rexpr = −I/(2(2b)^(m/2)) ∗ Exp[(c^2 − a^2)/(8b)] ∗ Gamma[m]
           (Exp[−Iac/(4b)] ∗ ParabolicCylinderD[−m, (c − Ia)/Sqrt[2b]]
            − Exp[Iac/(4b)] ∗ ParabolicCylinderD[−m, (c + Ia)/Sqrt[2b]]);
```

A short look at the expressions involved in this identity suggests to act on a, b, c with partial derivative, and on m with the shift operator. Observe that the identity holds for $a = 0$ as well. Annihilating ideals for both sides are readily computed, but unfortunately, they do not agree.

```
In[43]:= lhs = Annihilator[
           Integrate[x^(m − 1) Exp[−cx − bx^2] Sin[ax], {x, 0, Infinity}],
           {S[m], Der[a], Der[b], Der[c]},
           Assumptions → Re[m] > −1 && Re[b] > 0 && a ⩾ 0]
```

Out[43]= $\{aD_a + 2bD_b + cD_c + m,\ S_m + D_c,\ D_c^2 + D_b,$
$4b^2D_b^2 + 4bcD_bD_c + (-a^2 + 4bm + 6b - c^2)D_b + (2cm + 2c)D_c + (m^2 + m)\}$

```
In[44]:= rhs = Annihilator[rexpr, {S[m], Der[a], Der[b], Der[c]}]
```

Out[44]= $\{aD_a + 2bD_b + cD_c + m,\ D_c^2 + D_b,$
$4b^2D_b^2 + 4bcD_bD_c + (-a^2 + 4bm + 6b - c^2)D_b + (2cm + 2c)D_c + (m^2 + m),\ S_m^2 + D_b\}$

At least, it turns out that the latter is a subideal of the previous one:

```
In[45]:= Length[UnderTheStaircase[#]]& /@ {lhs, rhs}
```

Out[45]= $\{4, 8\}$

```
In[46]:= OreReduce[rhs, lhs]
```

Out[46]= $\{0, 0, 0, 0\}$

Why are there less relations found for the right-hand side? The reason is the factor $\sqrt{b}$ that appears in the argument of the parabolic cylinder function. **Annihilator** computes operators with coefficients that are polynomials in the variables corresponding to the Ore operators in the algebra. This can increase the order of the resulting operators, as the following two examples demonstrate:

```
In[47]:= Annihilator[ParabolicCylinderD[−m, Sqrt[b]], S[m]]
```

Out[47]= $\{(m+1)S_m^2 + \sqrt{b}S_m - 1\}$

```
In[48]:= Annihilator[ParabolicCylinderD[−m, Sqrt[b]], {Der[b], S[m]},
                     MonomialOrder → Lexicographic]
```

Out[48]= $\{(m^2 + 5m + 6)S_m^4 + (-b - 2m - 3)S_m^2 + 1,\ 4bD_b + (-2m^2 - 2m)S_m^2 + (b + 2m)\}$

However, the structure of our identity in question is special in the sense, that if computed with $\sqrt{b}$ in the coefficients, then all occurrences of it would disappear in the final result. This can be reproduced by introducing a new variable $s = \sqrt{b}$ in the original expression, and in the end the corresponding closure property "substitution" is performed. This produces the same annihilating ideal as for the left side:

In[49]:= **rexpr1 = Simplify[rexpr /. b → s^2, s > 0];**

In[50]:= **rhs1 = Annihilator[rexpr1, {S[m], Der[a], Der[s], Der[c]}]**

Out[50]= $\{aD_a + sD_s + cD_c + m,\ S_m + D_c,\ 2sD_c^2 + D_s,$
$2s^3D_s^2 + 4cs^2D_sD_c + (-a^2 - c^2 + 4ms^2 + 4s^2)D_s + (4cms + 4cs)D_c + (2m^2s + 2ms)\}$

In[51]:= **rhs1 = DFiniteSubstitute[rhs1, {s → Sqrt[b]},**
Algebra → OreAlgebra[S[m], Der[a], Der[b], Der[c]]]

Out[51]= $\{aD_a + 2bD_b + cD_c + m,\ S_m + D_c,\ D_c^2 + D_b,$
$4b^2D_b^2 + 4bcD_bD_c + (-a^2 + 4bm + 6b - c^2)D_b + (2cm + 2c)D_c + (m^2 + m)\}$

In[52]:= **GBEqual[lhs, rhs1]**

Out[52]= True

Alternatively, the operator D_b can be set aside. Since in this problem, the initial values to be checked are simple enough with symbolic m and b, also the shift in m will not be included. Considering m as a parameter has the additional advantage that the proof is valid for any m and not only for integer values.

In[53]:= **lhs = Annihilator[**
Integrate[x^(m − 1) Exp[−cx − bx^2] Sin[ax], {x, 0, Infinity}],
{Der[a], Der[c]}, Assumptions → Re[m] > −1 && Re[b] > 0 && a ⩾ 0]

Out[53]= $\{2bD_c^2 - aD_a - cD_c - m, 2bD_a^2 + aD_a + cD_c + m\}$

In[54]:= **rhs = Annihilator[rexpr, {Der[a], Der[c]}]**

Out[54]= $\{2bD_c^2 - aD_a - cD_c - m, 2bD_a^2 + aD_a + cD_c + m\}$

As already mentioned, the leading monomials in the Gröbner basis indicate the initial values to be compared. The integrals that remain to be computed now are simpler than the original one.

In[55]:= **uts = UnderTheStaircase[lhs]**

Out[55]= $\{1, D_c, D_a, D_aD_c\}$

In[56]:= **ApplyOreOperator[uts, x^(m − 1) Exp[−cx − bx^2] Sin[ax]]**
/. {a → 0, c → 0}

Out[56]= $\left\{0, 0, x^m e^{-bx^2}, x^{m+1}\left(-e^{-bx^2}\right)\right\}$

In[57]:= **Integrate[%, {x, 0, Infinity}, Assumptions → Re[m] > −1 && Re[b] > 0]**

Out[57]= $\left\{0, 0, \frac{1}{2}b^{-\frac{m}{2}-\frac{1}{2}}\text{Gamma}\left[\frac{m+1}{2}\right], -\frac{1}{2}b^{-\frac{m}{2}-1}\text{Gamma}\left[\frac{m}{2}+1\right]\right\}$

In[58]:= **FullSimplify[ApplyOreOperator[uts, rexpr] /. {a → 0, c → 0}]**

Out[58]= $\left\{0, 0, \frac{1}{2}b^{-\frac{m}{2}-\frac{1}{2}}\text{Gamma}\left[\frac{m+1}{2}\right], -\frac{1}{2}b^{-\frac{m}{2}-1}\text{Gamma}\left[\frac{m}{2}+1\right]\right\}.$

10. An elementary trigonometric integral

The final example is entry **4.535.1** in [**7**]:

$$\int_0^1 \frac{\arctan px}{1+p^2x}\,dx = \frac{1}{2p^2}\arctan p\,\ln\left(1+p^2\right). \tag{10.1}$$

In the computation of a differential equation for each side of the identity, an overshoot concerning the order is observed for the left-hand side (for brevity, only the leading monomials of both operators are displayed).

In[59]:= $\mathbf{lhs = Annihilator\big[Integrate[ArcTan[}\boldsymbol{px}\mathbf{]/(1+}\boldsymbol{xp}\mathbf{\hat{}2), \{}\boldsymbol{x}\mathbf{, 0, 1\}], Der[}\boldsymbol{p}\mathbf{]\big];}$

In[60]:= $\mathbf{rhs = Annihilator\big[(1/(2}\boldsymbol{p}\mathbf{\hat{}2))\ ArcTan[}\boldsymbol{p}\mathbf{]\ Log[1+}\boldsymbol{p}\mathbf{\hat{}2], Der[}\boldsymbol{p}\mathbf{]\big];}$

In[61]:= $\mathbf{LeadingPowerProduct\ /@\ Flatten[\{lhs, rhs\}]}$

Out[61]= $\{D_p^5, D_p^4\}$

It turns out that the 5th-order differential equation is a left multiple of the other one. An explanation for this non-agreement is desirable. This is obtained by considering the inhomogeneous part that remains after creative telescoping:

In[62]:= $\mathbf{\{\{op\}, \{inh\}\} = Annihilator\big[Integrate[ArcTan[}\boldsymbol{px}\mathbf{]/(1+}\boldsymbol{xp}\mathbf{\hat{}2), \{}\boldsymbol{x}\mathbf{, 0, 1\}],}$
$\mathbf{Der[}\boldsymbol{p}\mathbf{], Inhomogeneous \to True\big]}$

Out[62]= $\Big\{\{(p^4+p^2)D_p^2+(6p^3+4p)D_p+(6p^2+2)\},$
$\Big\{\frac{(-p^7-3p^5-3p^3-p)\mathrm{ArcTan}[p]-p^6-p^4+p^2+1}{p\,(p^2+1)^3}-\frac{1}{p}\Big\}\Big\}$

In[63]:= $\mathbf{FullSimplify[inh]}$

Out[63]= $-\mathrm{ArcTan}[p]-\frac{2p}{p^2+1}$

Hence the inhomogeneous differential equation that remains after telescoping is

$$\left(p^4+p^2\right)f''(p)+\left(6p^3+4p\right)f'(p)+\left(6p^2+2\right)f(p)=\frac{2p}{p^2+1}+\arctan p \tag{10.2}$$

which has to be homogenized. For this purpose, an annihilating operator for the inhomogeneous part (i.e., the right side of (10.2)) is computed and then left-multiplied to the operator **op**. By default, the closure property "addition" is used for such expressions. But a careful inspection shows that it can also be written as an operator application:

$$(2pD_p+1)(\arctan p)=\frac{2p}{p^2+1}+\arctan p. \tag{10.3}$$

In such situations the latter closure property is preferable as the following computations demonstrate:

In[64]:= $\mathbf{Annihilator\big[2(}\boldsymbol{p}\mathbf{/(1+}\boldsymbol{p}\mathbf{\hat{}2))+ArcTan[}\boldsymbol{p}\mathbf{], Der[}\boldsymbol{p}\mathbf{]\big]}$

Out[64]= $\big\{(p^5+2p^3+p)D_p^3+(7p^4+6p^2-1)D_p^2+8p^3D_p\big\}$

In[65]:= $\mathbf{Annihilator\big[ApplyOreOperator[2}\boldsymbol{p}\mathbf{\,Der[}\boldsymbol{p}\mathbf{]+1, ArcTan[}\boldsymbol{p}\mathbf{]], Der[}\boldsymbol{p}\mathbf{]\big]}$

Out[65]= $\big\{(p^4-2p^2-3)D_p^2+(2p^3-14p)D_p\big\}$

This leads to the same fourth-order differential equation previously obtained for the right-hand side:

In[66]:= **First[%] ** op**

Out[66]= $\left(p^8-p^6-5p^4-3p^2\right)D_p^4+\left(16p^7-32p^5-72p^3-24p\right)D_p^3$
$+\left(74p^6-224p^4-270p^2-36\right)D_p^2+\left(108p^5-444p^3-264p\right)D_p+\left(36p^4-192p^2-36\right)$

In[67]:= **GBEqual[%, rhs]**

Out[67]= True

At this point, the known evaluation of the integral is ignored, and the differential equation is employed to find it. The Mathematica command **DSolve** delivers the following four independent solutions:

In[68]:= **DSolve[ApplyOreOperator[First[rhs], *f*[*p*]] == 0, *f*[*p*], *p*][[1, 1, 2]]**

Out[68]= $\dfrac{\mathrm{C}[1]}{p^2}+\dfrac{\mathrm{C}[2]\,\mathrm{ArcTan}[p]}{p^2}+\dfrac{\mathrm{C}[3]\,\mathrm{Log}\left[p^2+1\right]}{2p^2}-\dfrac{\mathrm{C}[4]\,\mathrm{ArcTan}[p]\,\mathrm{Log}\left[p^2+1\right]}{6p^2}$

Four constants need to be determined. From the fact that the integral is 0 for $p=0$, the first two solutions are excluded, since they tend to infinity as $p\to 0$. The fourth solution tends to 0 for $p\to 0$, but the third one does not; hence C[3] must also vanish. The last constant can be determined from the first derivative with respect to p, evaluated at $p=0$:

$$\int_0^1 \frac{d}{dp}\left(\frac{\arctan px}{1+p^2x}\right)\bigg|_{p=0} dx=\int_0^1 x\,dx=\frac{1}{2}.$$

The remaining constant C[4] is seen to be 3 and the evaluation (10.1) has been rediscovered.

Acknowledgments. The first author was supported by NFS-DMS 0713836 as a postdoctoral fellow, and by grant P20162 of the Austrian FWF. The second author acknowledges the partial support of NSF-DMS 0713836.

References

[1] M. Abramowitz and I. Stegun. *Handbook of Mathematical Functions with Formulas, Graphs and Mathematical Tables*. Dover, New York, 1972.

[2] G. Almkvist and D. Zeilberger. The method of differentiating under the integral sign. *Jour. Symb. Comp.*, 10:571–591, 1990.

[3] F. Chyzak. *Fonctions holonomes en calcul formel*. PhD thesis, École polytechnique, 1998.

[4] F. Chyzak. An extension of Zeilberger's fast algorithm to general holonomic functions. *Discrete Mathematics*, 217(1-3):115–134, 2000.

[5] P. Flajolet, S. Gerhold, and B. Salvy. On the non-holonomic character of logarithms, powers, and the nth prime function. *Elec. Jour. Comb.*, 11:#A2, 2005.

[6] S. Gerhold. On some non-holonomic sequences. *Elec. Jour. Comb.*, 11:#R87, 2004.

[7] I. S. Gradshteyn and I. M. Ryzhik. *Table of Integrals, Series, and Products*. Edited by A. Jeffrey and D. Zwillinger. Academic Press, New York, 7th edition, 2007.

[8] C. Koutschan. *Advanced Applications of the Holonomic Systems Approach*. PhD thesis, RISC, Johannes Kepler University, Linz, Austria, 2009.

[9] Christoph Koutschan. HolonomicFunctions (User's Guide). Technical Report 10-01, RISC Report Series, Johannes Kepler University Linz, 2010. http://www.risc.uni-linz.ac.at/research/combinat/software/HolonomicFunctions/.

[10] V. Moll. The integrals in Gradshteyn and Ryzhik. Part 1: A family of logarithmic integrals. *Scientia*, 14:1–6, 2007.

[11] N. Takayama. An algorithm of constructing the integral of a module—an infinite dimensional analog of Gröbner basis. *ISSAC'90: Proceedings of the International Symposium on Symbolic and Algebraic Computation*, pages 206–211, 1990.

[12] D. Zeilberger. A holonomic systems approach to special function identities. *Journal of Computational and Applied Mathematics*, 32(3):321–368, 1990.

Received 26 04 2010, revised 21 08 2010

[a]Department of Mathematics,
Tulane University,
New Orleans,
LA 70118,
USA

E-mail address: koutschan@risc.uni-linz.ac.at
E-mail address: vhm@math.tulane.edu